Lesson 24: Taylor series

```
> restart;
```

Finding Taylor series

As we saw last time, Maple has the **taylor** command to find a given number of terms of a Taylor series of an expression.

```
> taylor((exp(x)-1-x)/x^2, x);
```

$$\frac{1}{2} + \frac{1}{6} x + \frac{1}{24} x^2 + \frac{1}{120} x^3 + \mathrm{O}(x^4) \tag{1.1}$$

But what if instead of the first few terms of the series, you want a formula for the whole series? This isn't always going to work, but sometimes it will:

```
> convert((exp(x)-1-x)/x^2, FormalPowerSeries, x);
```

$$\sum_{k=0}^{\infty} \frac{x^k}{(k+2)!} \tag{1.2}$$

```
> value(%);
```

$$\frac{e^x - 1 - x}{x^2} \tag{1.3}$$

If you want a series about some other point, say $x = a$:

```
> convert(exp(x), FormalPowerSeries, x=a);
```

$$\sum_{k=0}^{\infty} \frac{e^a (x-a)^k}{k!} \tag{1.4}$$

```
> convert(exp(x+x^2), FormalPowerSeries,x);
```

$$e^{x+x^2} \tag{1.5}$$

Maple doesn't know a formula for the coefficients of the Maclaurin series of e^{x+x^2}. Well, there isn't a very nice formula for them, as far as I know.

```
> taylor(exp(x+x^2),x,10);
```

$$1 + x + \frac{3}{2} x^2 + \frac{7}{6} x^3 + \frac{25}{24} x^4 + \frac{27}{40} x^5 + \frac{331}{720} x^6 + \frac{1303}{5040} x^7 + \frac{1979}{13440} x^8 + \frac{5357}{72576} x^9 + \mathrm{O}(x^{10}) \tag{1.6}$$

Convergence of Taylor series

Let's look at the Maclaurin series for the function e^x and its convergence to e^x. It's convenient to define a function that will calculate the n'th degree Maclaurin polynomial at a given point.

```
> P:= (n,t) -> eval( convert(taylor(exp(x), x=0, n+1),
  polynom), x=t);
```

$$P := (n, t) \to \mathit{convert}\left(\mathit{taylor}\left(e^x, x = 0, 1 + n\right), \mathit{polynom}\right)\Big|_{x = t}$$

```
> P(4,t); P(4,2);
```

$$1+t+\frac{1}{2}\,t^2+\frac{1}{6}\,t^3+\frac{1}{24}\,t^4$$

$$7$$

```
> plot([ exp(x), seq(P(n, x), n=1 .. 10)],
>        x=-6 .. 2, y=-2 .. 8);
```



Especially for $x > 0$, it's more informative to look at the difference between e^x and the Maclaurin polynomial.

```
> plot([seq(exp(x)-P(n, x), n=1 .. 12)],
>      x = -6 .. 6, y = -3 .. 3);
```



An animation is another possibility. I'm not using **animate** here because premature evaluation would cause trouble.

```
> with(plots):
  display([seq(plot(exp(x)-P(n,x),x=-7..7,y=-3..3,
    title=('n'=n)), n=1..16)], insequence=true);
```



It's almost (but not quite) true that the curves for x > 0 march off to the right at a constant rate of e^{-1} per step.

```
> display([seq(plot(exp(x+n*exp(-1)) - P(n, x+n*exp(-1)),
  x=-1 .. 1.5, y = -1 .. 2,title=('n'=n)),n=1..50)],insequence=
  true);
```



We'd get a very different picture for e.g. $\arctan(x)$. Here's the Taylor series:

```
> convert(arctan(x),FormalPowerSeries,x);
```

$$\sum_{k=0}^{\infty} \frac{(-1)^k x^{2k+1}}{2k+1} \tag{2.1}$$

```
> P := (n,t) -> subs(x=t,convert(taylor(arctan(x), x=0, n+1),
  polynom));
```

$$P := (n, t) \rightarrow subs(x = t, convert(taylor(\arctan(x), x = 0, 1 + n), polynom))$$

```
> P(7,t);
```

$$t - \frac{1}{3} t^3 + \frac{1}{5} t^5 - \frac{1}{7} t^7$$

```
> P(8,t);
```

$$t - \frac{1}{3} t^3 + \frac{1}{5} t^5 - \frac{1}{7} t^7$$

```
> plot([seq(arctan(x)-P(2*n,x), n=1..12)], x=-2 .. 2,
  y = -3 .. 3);
```




```
> display([seq(plot(arctan(x)-P(2*n,x),x = 0.7 .. 1.3, y = -3 .
  . 3,title=('Order'=2*n)),
  n = 1..30)], insequence=true);
```



The difference here is that the radius of convergence for arctan is 1, while for exp it is ∞. Outside the interval [-1,1], the series for arctan is useless for approximating $\arctan(x)$.

Manipulation with series

Various operations can be done to obtain new series from old series:
the basic operations of arithmetic, as well as substitution, differentiation and integration.

Example 1:

Starting with series any good Math 101 student should know, obtain the degree 10 Maclaurin polynomial for $\ln(1+x^2)\,\sin(\cos(x))$.

For example, the series for $\frac{1}{1-t}$ is a geometric series.

```
> s1 := 1/(1-t) = convert(taylor(1/(1-t), t=0, 10),polynom);
```

$$s1 := \frac{1}{1-t} = 1 + t + t^2 + t^3 + t^4 + t^5 + t^6 + t^7 + t^8 + t^9$$

Of course that's not literally true, it's just the first part of the series. But for some of the manipulations to work, I want a polynomial rather than a series.

Integrate this term-by-term and you have the series for $-\ln(1-t)$. Note that the constants of integration are the same, because both sides are 0 at $t=0$.

```
> s2 := int(lhs(s1),t) = int(rhs(s1),t);
```

$$s2 := -\ln(1-t) = t + \frac{1}{2}t^2 + \frac{1}{3}t^3 + \frac{1}{4}t^4 + \frac{1}{5}t^5 + \frac{1}{6}t^6 + \frac{1}{7}t^7 + \frac{1}{8}t^8 + \frac{1}{9}t^9 + \frac{1}{10}t^{10}$$

Change signs and substitute $t = -x^2$ and you have the series for $\ln(1+x^2)$.

```
> s3 := eval(-s2, t=-x^2);
```

$$s3 := \ln(1+x^2) = x^2 - \frac{1}{2}x^4 + \frac{1}{3}x^6 - \frac{1}{4}x^8 + \frac{1}{5}x^{10} - \frac{1}{6}x^{12} + \frac{1}{7}x^{14} - \frac{1}{8}x^{16} + \frac{1}{9}x^{18} - \frac{1}{10}x^{20}$$

That's way more terms than we need, we only want a degree-10 polynomial.

```
> s4 := lhs(s3) = convert(taylor(rhs(s3),x,11),polynom);
```

$$s4 := \ln(1+x^2) = x^2 - \frac{1}{2}x^4 + \frac{1}{3}x^6 - \frac{1}{4}x^8 + \frac{1}{5}x^{10}$$

That's one of the factors. Now for the $\sin(\cos(x))$. The Maclaurin series for cos(x) and sin(t) are also "known".

```
> convert(cos(x),FormalPowerSeries,x);
  convert(sin(x),FormalPowerSeries,x);
```

$$\sum_{k=0}^{\infty} \frac{(-1)^k x^{2k}}{(2k)!}$$

$$\sum_{k=0}^{\infty} \frac{(-1)^k x^{2k+1}}{(2k+1)!} \qquad \textbf{(3.1.1)}$$

```
> s5 := cos(x) = convert(taylor(cos(x),x,11),polynom);
```

$$s5 := \cos(x) = 1 - \frac{1}{2}x^2 + \frac{1}{24}x^4 - \frac{1}{720}x^6 + \frac{1}{40320}x^8 - \frac{1}{3628800}x^{10}$$

```
> s6 := sin(x) = convert(taylor(sin(x),x,11),polynom);
```

$$s6 := \sin(x) = x - \frac{1}{6}x^3 + \frac{1}{120}x^5 - \frac{1}{5040}x^7 + \frac{1}{362880}x^9$$

Now I want $\sin(\cos(x))$, which is approximately $\sin(s5)$.
But it would be wrong to use the Taylor series of $\sin(t)$ about $t=0$. When x is small, $\cos(x)$ is near 1, not near 0. So we want a series for $\sin(t)$ about $t=1$. A trigonometric identity relates $\sin(t)$ to sin and cos of $t-1$:

```
> trigident:= eval(sin(1+s)=expand(sin(1+s)),s=t-1);
```

$$trigident := \sin(t) = \sin(1)\cos(t-1) + \cos(1)\sin(t-1)$$

```
> s7 := eval(s5,x=t-1);
```

$$s7 := \cos(t-1) = 1 - \frac{1}{2}(t-1)^2 + \frac{1}{24}(t-1)^4 - \frac{1}{720}(t-1)^6 + \frac{1}{40320}(t-1)^8$$

$$-\frac{1}{3628800}(t-1)^{10}$$

```
> s8 := eval(s6,x=t-1);
```

$$s8 := \sin(t-1) = t-1-\frac{1}{6}(t-1)^3+\frac{1}{120}(t-1)^5-\frac{1}{5040}(t-1)^7+\frac{1}{362880}(t-1)^9$$

```
> s9 := eval(trigident,{s7,s8});
```

$$s9 := \sin(t) = \sin(1)\left(1-\frac{1}{2}(t-1)^2+\frac{1}{24}(t-1)^4-\frac{1}{720}(t-1)^6+\frac{1}{40320}(t-1)^8-\frac{1}{3628800}(t-1)^{10}\right)+\cos(1)\left(t-1-\frac{1}{6}(t-1)^3+\frac{1}{120}(t-1)^5-\frac{1}{5040}(t-1)^7+\frac{1}{362880}(t-1)^9\right)$$

```
> s10 := sin(t) = convert(taylor(rhs(s9),t=1,11),polynom);
```

$$s10 := \sin(t) = \sin(1)+\cos(1)(t-1)-\frac{1}{2}\sin(1)(t-1)^2-\frac{1}{6}\cos(1)(t-1)^3+\frac{1}{24}\sin(1)(t-1)^4+\frac{1}{120}\cos(1)(t-1)^5-\frac{1}{720}\sin(1)(t-1)^6-\frac{1}{5040}\cos(1)(t-1)^7+\frac{1}{40320}\sin(1)(t-1)^8+\frac{1}{362880}\cos(1)(t-1)^9-\frac{1}{3628800}\sin(1)(t-1)^{10}$$

```
> s11 := eval(lhs(s10),t=cos(x)) = eval(rhs(s10),t = rhs(s5))
;
```

$$s11 := \sin(\cos(x)) = \sin(1)+\cos(1)\left(-\frac{1}{2}x^2+\frac{1}{24}x^4-\frac{1}{720}x^6+\frac{1}{40320}x^8-\frac{1}{3628800}x^{10}\right)-\frac{1}{2}\sin(1)\left(-\frac{1}{2}x^2+\frac{1}{24}x^4-\frac{1}{720}x^6+\frac{1}{40320}x^8-\frac{1}{3628800}x^{10}\right)^2-\frac{1}{6}\cos(1)\left(-\frac{1}{2}x^2+\frac{1}{24}x^4-\frac{1}{720}x^6+\frac{1}{40320}x^8-\frac{1}{3628800}x^{10}\right)^3+\frac{1}{24}\sin(1)\left(-\frac{1}{2}x^2+\frac{1}{24}x^4-\frac{1}{720}x^6+\frac{1}{40320}x^8-\frac{1}{3628800}x^{10}\right)^4+\frac{1}{120}\cos(1)\left(-\frac{1}{2}x^2+\frac{1}{24}x^4-\frac{1}{720}x^6+\frac{1}{40320}x^8-\frac{1}{3628800}x^{10}\right)^5-\frac{1}{720}\sin(1)\left(-\frac{1}{2}x^2+\frac{1}{24}x^4-\frac{1}{720}x^6+\frac{1}{40320}x^8-\frac{1}{3628800}x^{10}\right)^6-\frac{1}{5040}\cos(1)\left(-\frac{1}{2}x^2+\frac{1}{24}x^4-\frac{1}{720}x^6+\frac{1}{40320}x^8-\frac{1}{3628800}x^{10}\right)^7+\frac{1}{40320}\sin(1)\left(-\frac{1}{2}x^2+\frac{1}{24}x^4-\frac{1}{720}x^6+\frac{1}{40320}x^8-\frac{1}{3628800}x^{10}\right)^8+\frac{1}{362880}\cos(1)\left(-\frac{1}{2}x^2+\frac{1}{24}x^4-\frac{1}{720}x^6\right.$$

$$+\frac{1}{40320}x^8-\frac{1}{3628800}x^{10}\Big)^9-\frac{1}{3628800}\sin(1)\left(-\frac{1}{2}x^2+\frac{1}{24}x^4-\frac{1}{720}x^6+\frac{1}{40320}x^8-\frac{1}{3628800}x^{10}\right)^{10}$$

```
> s12 := lhs(%) = convert(taylor(rhs(%),x,11),polynom);
```

$$s12 := \sin(\cos(x)) = \sin(1)-\frac{1}{2}\cos(1)x^2+\left(\frac{1}{24}\cos(1)-\frac{1}{8}\sin(1)\right)x^4+\left(\frac{7}{360}\cos(1)+\frac{1}{48}\sin(1)\right)x^6+\left(-\frac{209}{40320}\cos(1)+\frac{1}{960}\sin(1)\right)x^8+\left(\frac{1259}{3628800}\cos(1)-\frac{193}{241920}\sin(1)\right)x^{10}$$

And finally:

```
> lhs(s4)*lhs(s12)=convert(taylor(rhs(s4)*rhs(s12),x,11),
  polynom);
```

$$\ln(1+x^2)\sin(\cos(x)) = \sin(1)x^2+\left(-\frac{1}{2}\cos(1)-\frac{1}{2}\sin(1)\right)x^4+\left(\frac{7}{24}\cos(1)+\frac{5}{24}\sin(1)\right)x^6+\left(-\frac{121}{720}\cos(1)-\frac{1}{6}\sin(1)\right)x^8+\left(\frac{143}{960}\sin(1)+\frac{4999}{40320}\cos(1)\right)x^{10}$$

Of course, we could have used one "**taylor**" command, this was just to see how it could be done.

```
> taylor(ln(1+x^2)*sin(cos(x)),x,11);
```

$$\sin(1)x^2+\left(-\frac{1}{2}\cos(1)-\frac{1}{2}\sin(1)\right)x^4+\left(\frac{7}{24}\cos(1)+\frac{5}{24}\sin(1)\right)x^6+\left(-\frac{121}{720}\cos(1)-\frac{1}{6}\sin(1)\right)x^8+\left(\frac{143}{960}\sin(1)+\frac{4999}{40320}\cos(1)\right)x^{10}+O(x^{12})$$

```
> taylor(rhs(%%)-%, x, 11);
```

$$O(x^{12}) \qquad (3.1.2)$$

Example 2:

Find the Taylor series for $f(x)$ about $x=0$ up to the x^6 term, if $y=f(x)$ satisfies the equation $(1+x)\,e^y-y^2\,e^x=1+x^2\,y$ with $f(0)=0$.

```
> eq:= (1+x)*exp(y)-y^2*exp(x)=1+ x^2*y;
```

$$eq := (1+x)\,e^y-y^2\,e^x=1+x^2\,y$$

Check that it works with $f(0)=0$.

```
> eval(eq,{x=0,y=0});
```

$$1=1$$

So we want the first 6 terms to look like this:

```
> yseries := add(a[n]*x^n,n=1..6);
```

$$yseries := a_1x+a_2x^2+a_3x^3+a_4x^4+a_5x^5+a_6x^6$$

Substitute this in to the difference of the two sides of the equation.

```
> eval(lhs(eq)-rhs(eq), y=yseries);
```

$$(1+x)\,e^{a_1 x + a_2 x^2 + a_3 x^3 + a_4 x^4 + a_5 x^5 + a_6 x^6} - \left(a_1 x + a_2 x^2 + a_3 x^3 + a_4 x^4 + a_5 x^5 + a_6 x^6\right)^2 e^x - 1 - x^2 \left(a_1 x + a_2 x^2 + a_3 x^3 + a_4 x^4 + a_5 x^5 + a_6 x^6\right)$$

```
> taylor(%, x, 7);
```

$$(a_1+1)\,x + \left(-\frac{1}{2} a_1^2 + a_2 + a_1\right) x^2 + \left(a_3 - a_1 a_2 + \frac{1}{6} a_1^3 + a_2 - \frac{1}{2} a_1^2 - a_1\right) x^3 + \left(a_3 - a_1 a_2 + \frac{1}{6} a_1^3 + a_4 - a_1 a_3 - \frac{1}{2} a_2^2 + \frac{1}{2} a_2 a_1^2 + \frac{1}{24} a_1^4 - \frac{1}{2} a_1^2 - a_2\right) x^4 + \left(a_5 - a_1 a_4 - a_3 a_2 + \frac{1}{2} a_3 a_1^2 + \frac{1}{2} a_1 a_2^2 + \frac{1}{6} a_2 a_1^3 + \frac{1}{120} a_1^5 + a_4 - a_1 a_3 - \frac{1}{2} a_2^2 + \frac{1}{2} a_2 a_1^2 + \frac{1}{24} a_1^4 - a_1 a_2 - \frac{1}{6} a_1^2 - a_3\right) x^5 + \left(a_3 a_1 a_2 - a_4 + a_5 + a_6 + \frac{1}{120} a_1^5 + \frac{1}{6} a_2^3 + \frac{1}{720} a_1^6 - \frac{1}{24} a_1^2 - \frac{1}{3} a_1 a_2 - a_1 a_3 - \frac{1}{2} a_2^2 - a_1 a_4 - a_3 a_2 + \frac{1}{2} a_3 a_1^2 + \frac{1}{2} a_1 a_2^2 + \frac{1}{6} a_2 a_1^3 - a_1 a_5 - a_4 a_2 + \frac{1}{2} a_4 a_1^2 - \frac{1}{2} a_3^2 + \frac{1}{6} a_3 a_1^3 + \frac{1}{4} a_2^2 a_1^2 + \frac{1}{24} a_2 a_1^4\right) x^6 + \mathrm{O}(x^7)$$

Now the coefficients of each power of x should match.

```
> equations:= {seq(coeff(%,x,n),n=1..6)};
```

$$equations := \left\{a_1 + 1,\ -\frac{1}{2} a_1^2 + a_2 + a_1,\ a_3 - a_1 a_2 + \frac{1}{6} a_1^3 + a_2 - \frac{1}{2} a_1^2 - a_1,\ a_3 - a_1 a_2 + \frac{1}{6} a_1^3 + a_4 - a_1 a_3 - \frac{1}{2} a_2^2 + \frac{1}{2} a_2 a_1^2 + \frac{1}{24} a_1^4 - \frac{1}{2} a_1^2 - a_2,\ a_5 - a_1 a_4 - a_3 a_2 + \frac{1}{2} a_3 a_1^2 + \frac{1}{2} a_1 a_2^2 + \frac{1}{6} a_2 a_1^3 + \frac{1}{120} a_1^5 + a_4 - a_1 a_3 - \frac{1}{2} a_2^2 + \frac{1}{2} a_2 a_1^2 + \frac{1}{24} a_1^4 - a_1 a_2 - \frac{1}{6} a_1^2 - a_3,\ a_3 a_1 a_2 - a_4 + a_5 + a_6 + \frac{1}{120} a_1^5 + \frac{1}{6} a_2^3 + \frac{1}{720} a_1^6 - \frac{1}{24} a_1^2 - \frac{1}{3} a_1 a_2 - a_1 a_3 - \frac{1}{2} a_2^2 - a_1 a_4 - a_3 a_2 + \frac{1}{2} a_3 a_1^2 + \frac{1}{2} a_1 a_2^2 + \frac{1}{6} a_2 a_1^3 - a_1 a_5 - a_4 a_2 + \frac{1}{2} a_4 a_1^2 - \frac{1}{2} a_3^2 + \frac{1}{6} a_3 a_1^3 + \frac{1}{4} a_2^2 a_1^2 + \frac{1}{24} a_2 a_1^4\right\}$$

```
> solve(equations);
```

$$\left\{a_1 = -1,\ a_2 = \frac{3}{2},\ a_3 = -\frac{10}{3},\ a_4 = \frac{23}{3},\ a_5 = -\frac{1097}{60},\ a_6 = \frac{8117}{180}\right\}$$

So here is our answer.

```
> answer:= eval( yseries, %);
```

$$answer := -x + \frac{3}{2} x^2 - \frac{10}{3} x^3 + \frac{23}{3} x^4 - \frac{1097}{60} x^5 + \frac{8117}{180} x^6$$

I'll check this graphically.

```
> with(plots):
```

```
P0:= implicitplot(eq,x=-1..1,y=-1..2,colour=blue):
for j from 1 to 6 do
  frame[j]:= display([P0,plot(convert(taylor(answer,x,j+1),
polynom),x=-1..1)],title=('Degree'=j))
end do:
display([seq(frame[j],j=1..6)],insequence=true,view=[-1..1,
-1..2]);
```



Here's a look at the curve on a larger scale.

```
> implicitplot(eq,x=-10..10,y=-10..10,gridrefine=3);
```



Rather than doing this with solve, you can use a version of Newton's method.

```
> f:= unapply(lhs(eq)-rhs(eq),(x,y));
```

$$f := (x, y) \rightarrow (1 + x)\, e^{y} - y^{2}\, e^{x} - 1 - x^{2}\, y$$

```
> newt:= (y,n) -> convert(normal(taylor(y-f(x,y)/D[2](f)(x,
  y), x, n)),polynom);
```

$$newt := (y, n) \rightarrow convert\left(normal\left(taylor\left(y - \frac{f(x, y)}{D_2(f)\,(x, y)}, x, n \right) \right), polynom \right)$$

```
> y1 := newt(0,2);
```

$$y1 := -x$$

```
> y2 := newt(y1,4);
```

$$y2 := -x + \frac{3}{2}\, x^{2} - \frac{10}{3}\, x^{3}$$

```
> normal(taylor(y2 - answer,x,4));
```

$$\mathrm{O}(x^{4})$$

How does this work? If $f(x, y_k) = \mathrm{O}(x^{k})$ and $\frac{\partial}{\partial y} f(x, y_k)$ has a nonzero limit as $x \rightarrow 0$, then I claim $f(x, y_{k+1}) = O(x^{2k})$ where $y_{k+1} = y_k - \dfrac{f(x, y_k)}{\frac{\partial}{\partial y} f(x, y_k)}$. In fact

$$f(x, y_{k+1}) = f(x, y_k) + \frac{\partial}{\partial y} f(x, y_k)\,(y_{k+1} - y_k) + O\left((y_{k+1} - y_k)^2\right)$$
$$= O\left((y_{k+1} - y_k)^2\right) \text{ and } y_{k+1} - y_k = O(x^k)$$

In other words, once you get an approximation that works to a certain order $O(x^k)$, each application of Newton's method will at least double the order of approximation.

Maple objects introduced in this lesson

convert(..., FormalPowerSeries, ...)